Phys 238: Driven Oscillations
Resonance

In[79]:=
```
Clear["Global`*"]; DateString[]
```
Out[79]=
```
Fri 27 Feb 2026 11:50:36
```

In[80]:=
```
SetDirectory[NotebookDirectory[]];
(* Find and save files alongside this notebook. *)
```

Differential Equation and Basic Parameters

The governing differential equation is:

$$\theta''[t] = -\omega 0^2\, \theta(t) - \gamma\, \theta'(t) + \alpha 0\, \mathrm{Sin}[\omega d\, t]$$

where κ is the restoring torque, I is the moment of inertia, $\omega 0 = \mathrm{Sqrt}[\kappa / I]$, γ = drag, and $\alpha 0 = \tau 0 / I$, where $\tau 0$ is the amplitude of the driving torque, ωd is the frequency of the driving. Related quantities are $Q = \omega 0/\gamma$ and linear frequency $f0 = \omega 0/2\pi$ and $fd = \omega d/2\,\pi$

$$\theta''[t] = -\omega 0^2\, \theta(t) - (\omega 0 / Q)\, \theta'(t) + \alpha 0\, \mathrm{Sin}[\omega d\, t]$$

Physical constants

In[81]:=
```
ω0 = 4.4 (* rad/s, typical for torsional oscillator with 2 brass quadrants. *);
f0 = ω0 / (2 π) (* Natural frequency in Hz. *)
tmax1 = 60 (* Run first simulation this long *);
```
Out[82]=
```
0.700282
```

Do the numerical integration

In[84]:=
```
damped = ParametricNDSolveValue[
  {θ''[t] == -(2 π f0)^2 θ[t] - (2 π f0 / Q) θ'[t] , θ[0] == 1, θ'[0] == 0},
   θ,
   {t, 0, tmax1}, {Q}]
```
Out[84]=
```
ParametricFunction[ 

]
```

Damped Oscillations -- the effect of Q

In[85]:=
```
Manipulate[Plot[damped[Q][t], {t, 0, tmax1},
  PlotRange → {-1, 1}, PlotPoints → 300, PlotLabel →
   StringForm["Q = ``, t1/2 = ``s", Q, NumberForm[2 Log[2] Q / ω0, {4, 2}]],
  LabelStyle → Larger, AxesLabel → {"t (s)", "Angle (rad)"},
  ImageSize → Scaled[0.8]],
 {{Q, 20}, 1, 40, 1, Appearance → "Open"}]
```

Out[85]=



Driven Oscillations -- initial transients

Set initial position to 0. Set a longer integration time to see transients.

In[86]:=
```
tmax2 = 90;
```

In[87]:=
```
driven = ParametricNDSolveValue[
  {θ''[t] == -(2 π f0)^2 θ[t] - (2 π f0 / Q) θ'[t] + α0 Sin[2 π fd t],
   θ[0] == 0, θ'[0] == 0},
   θ,
   {t, 0, tmax2}, {Q, α0, fd}]
```

Out[87]=

ParametricFunction[Expression: θ
Parameters: {Q, α0, fd}]

In[88]:=
```
Manipulate[Plot[driven[Q, α0, fd][t], {t, 0, tmax2}, PlotPoints → 300,
  PlotRange → {-8, 8}, PlotLabel → StringForm["Q = ``", Q], LabelStyle → Larger,
  AxesLabel → {"t (s)", "Angle (rad)"}, ImageSize → Scaled[0.8]],
 {{Q, 30}, 1, 40, 1, Appearance → "Labeled"},
 {{α0, 5}, 1, 20, 1, Appearance → "Labeled"},
 {{fd, 0.6}, 0.2, 2, 0.02, Appearance → "Open"}]
```

Out[88]=



Steady State Response -- Resonance

In[89]:=
```
resonance[α0_, f0_, f_, Q_] := α0 / Sqrt[(f0^2 - f^2)^2 + (f0 f / Q)^2]
```

In[90]:=
```
Manipulate[Plot[resonance[α0, f0, fd, Q], {fd, 0, 2 f0}, PlotPoints → 300,
  PlotRange → All, PlotLabel → StringForm["Q = ``", Q], LabelStyle → Larger,
  AxesLabel → {"fd (Hz)", "Amplitude"}, ImageSize → Scaled[0.8]],
 {{Q, 30}, 1, 40, 1, Appearance → "Labeled"},
 {{α0, 5}, 1, 20, 1, Appearance → "Labeled"}]
```

Out[90]=



In[91]:=
```
Plot[{resonance[1, f0, fd, 5],
  resonance[1, f0, fd, 10],
  resonance[1, f0, fd, 15]}, {fd, 0.4, 1.0}, PlotPoints → 300, PlotRange → All,
 PlotLegends → {"Q = 5", "Q = 10", "Q = 15"}, LabelStyle → Larger,
 AxesLabel → {"fd (Hz)", "Amplitude"}, ImageSize → Scaled[0.8]]
```

Out[91]=



In[92]:=
```
phase[f0_, f_, Q_] := ArcTan[ f0^2 - f^2 , f0 f / Q]
```

In[93]:=
```
Manipulate[Plot[{phase[f0, fd, Q], π}, {fd, 0, 2 f0}, PlotPoints → 300,
  PlotRange → {0, π}, PlotLabel → StringForm["Q = ``", Q],
  PlotLegends → {"Phase", "π"}, LabelStyle → Larger,
  AxesLabel → {"fd (Hz)", "Phase"}, ImageSize → Scaled[0.8]],
 {{Q, 5}, 1, 40, 1, Appearance → "Open"}]
```

Out[93]=



In[94]:=
```
Plot[{phase[f0, fd, 5],
  phase[f0, fd, 10],
  phase[f0, fd, 15]}, {fd, 0.4, 1.0}, PlotPoints → 300, PlotRange → All,
 PlotLegends → {"Q = 5", "Q = 10", "Q = 15"}, LabelStyle → Larger,
 AxesLabel → {"fd (Hz)", "Phase (radians)"}, ImageSize → Scaled[0.8]]
```

Out[94]=